
Apéndice A

Manual de usuario de la BT6811

A.1 Introducción

La BT6811 esuna tarjeta microcontroladora basadaen el MC68hc11 de Motor ola. Su obje-
tivo escontrolar de forma autónoma cuatro servomecanismos,permitiendo una conexión
sencilla de los mismos y facilitando su alimentación.

Paradesarrollar estaplaca seha partido del modelo CT6811, anteriormente desarrolla-
do y se han eliminado los componentes que no se van a utilizar . Al �nal se ha obtenido
una placa con unas dimensiones de 48 mm x 58 mm.

Los elementos que se han eliminado con relación a su antecesorahan sido: el dispo-
sitivo de comunicaciones MAX232, los puertos B, C, Control, el jack de alimentación, los
switches de con�guración y algunos jumpers de con�guración. Por contra, se han intr o-
ducido los siguientes elementos: Una clema para alimentar los motores, un jumper para
poner la alimentación de los motores interna, cuatro conexionesdir ectaspara servomeca-
nismos y un puerto mixto, que sacaal exterior el puerto C y el E.

Seha conseguido rutar la placa a una sola cara, logrando así abaratar los costesde fa-
bricación y permitir que la placa sepueda construir artesanalmente.En el casode soldar a
mano sehará por la cara de atrás y la tarjeta no necesitarátaladros metalizados. Una con-
secuenciade esto,esque sehan perdido algunas compatibilidades de puertos con relación
a la CT6811, pero esto no es importante debido a la nueva función que ha de desarrollar
estaplaca. En los capítulos siguientes seirán detallando todas las funciones.

A.2 Modos de funcionamiento

Realmente la BT6811 está pensada para funcionar en modo autónomo, dentro de esta
modalidad sepueden distinguir dos tipos de modos.

1. Arrancando en Bootstrap (jumper JP1puesto): en este caso al alimentar la BT6811
se ejecuta un programa interno del MC6811 que se queda esperando a recibir un
programa por el puerto serie. Sepuede utilizar esto para volcar programas desde
el PC 1, o desde cualquier otro dispositivo serie, como por ejemplo una calculadora

1Utilizando un adaptador especialya que la BT6811 no lleva MAX y por lo tanto no puede recibir señales
del PC.

Figura A.1: Red de tarjetas BT6811.

HP. Para pasar al modo autónomo desde el bootstraphay que colocar el jumper JP2.
Así, cada vez que se haga un reset se le indica al programa bootstrapque ejecuteel
programa grabado en la EEPROM interna. El inconveniente de estemétodo es que
al poner el jumper JP2se pierden las comunicaciones serie asíncronas (SCI) de la
BT6811.

2. Arrancando en Single-Chip (jumper JP1quitado): en este caso al intr oducir la ali-
mentación en la tarjeta se ejecuta el programa apuntado por el vector de reset. De-
bido a que este vector se encuentra en la posición $FFFF, será necesario disponer
de un microcontrolador modelo E2 (que tiene 2K de EEPROM situados en el tramo
$F800-$FFFF).Al grabar el programa en la EEPROM hay que asegurarsede poner el
vector de resetapuntando al principio del programa. Con estemétodo no secoloca
el jumper JP2y no sepierden las comunicacionesserie asíncronas.

En ambos casos,se ha pensado en la posibilidad de montar una red de tarjetas BT6811.
Paraello sepuede utilizar el SPIo el SCI,cuyos pines estánen el PUERTO D. La red estará
formada por un conjunto de BT6811 esclavasy un dispositivo maestro, como por ejemplo
una CT6811. Con la estructura planteada se puede tener un PC comunicándose con una
CT6811 por el puerto serie y éstaa su vez controlando la red de BT6811 por el puerto SPI.
Ver �gura A.1

Otra necesidaddifer ente espoder cargar o grabar programas desde el PC en el micro-
controlador 6811 de la BT6811 sin tener que quitarlo del zócalo. Es decir usar la BT6811
como entrenadora para ejecutar el “downmcu”, el “ctdialog” o el “mcboot”. Como la
tarjeta no lleva un MAX232 habrá que proporcionarle uno, las alternativas son dos: o se
construye un adaptador o se utiliza una tarjeta CT6811 que ya lo lleva incorporado. En
ambos casosla conexión del adaptador sehará por el puerto D de la BT6811.

En estemanual se hace referencia al segundo caso,es decir a utilizar la CT6811 como
interfaz entre el PC y la placa BT6811. Lo que sehará esconectar los puertos D de ambas
tarjetas con un cablede bus estándar, quitar el microcontrolador de la CT6811 y alimentar
con 5 voltios ambastarjetas. El PC seconectamediante el cabletelefónico al conector de la
CT6811. A partir de estemomento sepueden ejecutar todos los programas citados antes.
La única difer encia es que ahora no hay “r esetsoftware”, por lo que habrá que pulsar el
resetde la BT6811 de forma manual. Salvoestepequeño detalle todo lo demássemantiene
como en la CT6811. Ver �gura A.2

Figura A.2: Esquemade conexión de la BT6811 en modo entrenadora.

Es importante que se quite el micro de la CT6811 para poder cargar los programas
en la BT6811. Si no se quita apareceráun con�icto entre los programas de BOOTSTRAP
de cada micro. Esto último se puede generalizar a una red de BT6811, si las tenemos
todas conectadasentre sí por un cable de BUS normal y arrancamos todos los micros en
BOOTSTRAPpara ser cargados desdeel adaptador del MAX, resultará que no funcionará
nada. Los programas bootstrapde cadaBT6811 interferirán la línea TX y RX corrompiendo
la información. Debido a esto cada BT6811 conectadaa una red habrá que reprogramarla
aisladamente.

A.3 Jumpers de con�guración de la BT6811

La BT6811 tiene seis jumpers de con�guración, se procede a describir la función de cada
uno de ellos. Ver �guraA.3

1. JP1(#B/S): Esel encargado de seleccionarel modo de funcionamiento del microcon-
trolador. Siel jumper estapuesto secon�gura el modo “bootstrap”, por el contrario si
estaquitado secon�gura el modo “single-chip”. Parapoder utilizar ésteúltimo será
necesariodisponer del vector de reseten EEPROM,como ocurre con los modelos de
microcontroladoresMC68hc11E2.El jumper estarápuesto.

2. JP2(Go EEPROM): Este jumper tiene sentido cuando el JP1estáconectado,esdecir
cuando la BT6811 esta con�gurada en modo “bootstrap”. Si se conectael JP2,cada
vez que se pulse “r eset” en la BT6811 se ejecutará el programa que esté grabado
en la EEPROM interna. Si por el contrario se deja quitado, al recibir el “r eset” se
ejecutaráel programa “bootstrap” y por lo tanto la placa estaráfuncionando en modo
entrenador, esperandoa recibir el programa a ejecutar por el puerto serie.

Figura A.3: Disposición de los jumpers en la BT6811.

3. JP3(LVI): Estejumper estaconectadoinicialmente. Su función esproteger el progra-
ma grabado en la EEPROM,haciendo un resetdel sistemacadavez que la alimenta-
ción bajede 4,5v. Por esosi seutilizan aplicaciones que inducen ruido eléctrico en la
alimentación, seráaconsejablequitar el jumper. Por ejemplo, esel casode utilizar la
placa para mover motoresestando con�gurada para alimentarlos internamente.

4. JP4(EXP):Su función espermitir sacarpor el puerto D la alimentación TTL. Esto es
muy útil a la hora de conectarotrosperiféricos a la BT6811. Inicialmente estáquitado,
debido a que el periférico que sesueleconectaresuna CT6811 y éstano puede recibir
ninguna señalde alimentación por dicho puerto.

5. JP5(Vcc ON): Su utilidad es permitir llevar la alimentación TTL a los motores, de
manera que no sea necesario conectar una fuente externa (batería, pila, etc.) para
poder moverlos. Inicialmente estaquitado, por lo que habrá que utilizar la segunda
fuente para poder mover los motores.

6. JP6(SS):ecto está puesto, su utilidad es facilitar el control en las redes de BT6811
mediante el SPI.Al quitar el jumper JP6,la línea SS(Slave Select)que proviene del
maestro no se conecta al SSdel esclavo. De esta forma podemos hacer que otros
pines del micro maestro sepuedan conectaral SSdel esclavo,pudiendo activar cada
esclavosin interferir en los demás.

A.4 Localización del resto de componentes

Ya se han descrito los jumpers de con�guración, ahora se describen el resto de compo-
nentesde la BT6811. Ver �gura A.4

� LED: Estaconectadoal bit_4 del puerto B. Estepuerto estamapeado en la dir ección
$1004del microcontrolador MC68hc11. Si sepone a uno dicho bit, entoncesseencen-
derá el LED.

Figura A.4: Restode componentesde la BT6811.

� Reset:Cadavez que sepulse el “pulsador ” seestaráhaciendo un resetdel MC68hc11.
El programa en ejecución se parará y volverá a inicializarse el sistema. El modo de
arranque seráel con�gurado por medio de los jumpers JP2y JP1.

� V_TTL: Esla clema por donde seintr oduce la tensión de alimentación TTL (5v) nece-
saria para la electrónica.

� V_Motor: Es la clema por donde se intr oduce la tensión de alimentación de los mo-
tores. Si seconectael jumper JP5seevita usar estaclema, al permitir que la tensión
TTL (la intr oducida por V_TTL) seala que alimente los motores.

� Puerto D: Es un bus de expansión, por él van las señalesde comunicaciones serie y
las del SPI.

� Puerto A: Esun bus de expansión, por él van las señalesdel puerto A del MC68hc11
junto con V_TTL y GND. Estepuerto estotalmente compatible con el de la CT6811.

� Puerto X: Es un bus de expansión mixto. Por él se sacanal exterior las señalesdel
puerto C y del puerto E del MC68hc11. El puerto E esde entrada mientras que el C
es bidir eccional, hay que tener en cuenta esto para evitar cortocircuitar dos salidas.
Por ejemplo si se quiere intr oducir una señal por el pin PX1, hay que con�gurar el
bit del puerto C correspondiente a PX1 como entrada. Luego se podrá leer el valor
bien por el puerto E o por el C. El estado predeterminado del Puerto C y del Puerto
E esde entrada, por lo que no habrá ningún con�icto en el momento de arranque del
microcontrolador.Ver tabla A.1

� FB1,FB2,FB3,FB4: Son jumpers triples donde se van a conectar los servomecanis-
mos. El FB1secontrola con el bit_0 del puerto B del MC68hc11,el FB2con el bit_1, el
FB3con el bit_2 y el FB4con el bit_3.

Puerto B PB7 PB6 PB5 PB4 PB3 PB2 PB1 PB0
$1004 – – – LED FB4 FB3 FB2 FB1

PUERTO X Puerto C (bidir eccional D) Puerto E (entada D/A)
PX0 PC0 PE3
PX1 PC1 PE7
PX2 PC2 PE2
PX3 PC3 PE6
PX4 PC4 PE5
PX5 PC5 PE1
PX6 PC6 PE4
PX7 PC7 PE0

Tabla A.1: Correspondenciade bits en el Puerto X.

� El restode componentessonel reloj y suscondensadoresasociados,el propio MC68hc11,
condensadoresde �ltrado de alimentación y resistenciasde polarización.

A.5 Pines de los buses de expansión

A.6 Utilización de la BT6811

La BT6811 se puede usar como entrenadora, o como producto �nal (autónomo), tanto en
modo aislado como en red. Su diseño está pensado para los modos autónomos y dentro
de ellos el de red. Para usarsecomo entrenadora sedebe disponer de un adaptador serie
como el max232, éste será externo y permitirá unir la tarjeta con el puerto serie del PC.
Como la CT6811 incorpora uno sepuede utilizar éstacomo adaptador.

En este apartado se van a describir mediante ejemplos todas las formas de uso. Se
empezarápor el modo entrenador y seterminará con el modo autónomo en red.

A.6.1 BT6811 trabajando como entrenadora

La utilidad de usar la BT6811 como entrenadora radica en la posibilidad de programar la
tarjeta sin necesidadde sacarel microcontrolador de su zócalo, detalle importante ya que
en algunas ocasionessetiene un difícil accesoa él. Como ya seha dicho con anterioridad,
para poder usar estemodo senecesitaun adaptador externo que permita conectarel puerto
serie de la BT6811 con el puerto serie del PC. En el mercado existen varios integrados que
realizan estafunción, uno de ellos esel MAX232, que esel que seha utilizado en la CT6811.
Por estarazón seha diseñado la BT6811 para poder ser conectadaal PC usando la CT6811
como adaptador. La única condición essacarel microcontrolador del zócalo de la CT6811
y unir ambas tarjetas por el puerto D mediante un cable bus estándar de diez hilos. En la
�gura A.2 sedescribe un esquemade conexión.

Paracon�gurar la BT6811 en modo entrenador hay que colocar los jumpers de la sigui-
ente forma:

1. Jumper JP1puesto para seleccionarel modo Bootstrap.

2. Jumper JP2quitado para permitir la ejecucióndel programa Bootstrap.

3. Los jumpers JP3,JP5y JP6no in�uyen en la con�guración del modo entrenador.

4. Jumper JP4desconectadoya que el modo auxiliar de la BT6811 serála CT6811.

Con la estructura y la con�guración planteadas la BT6811 secomporta como una CT6811
pero sin la función de reset software. Esto hace que la mayoría de los procedimientos y
ejemplos que sedescriben en el manual de la CT6811 seanválidos para la BT6811. A pesar
de esto se describe a continuación cómo enviar programas a la tarjeta y cómo dejarlos
autónomos, pero conviene recordar que se pueden obtener ejemplos más prácticos en el
manual de la CT6811.

Para descargar un programa desde el PC se pueden utilizar los programas downmcu,
ctload y mcboot. En este caso se va a emplear el downmcupor ser el más conocido y
disponible para las plataformas MsDos y Linux. Ahora hay que mencionar una difer -
encia fundamental con respectoal procesode carga con la entrenadora CT6811. En dicha
entrenadora existe la función de reset software que permite que el PC pueda resetearal
microcontrolador justo antesdel procesode carga. En la BT6811 esafunción no existe por
lo que el procesode reset tendrá que ser manual. El método para ello es pulsar el botón
de reset de la BT6811 e inmediatamente después pulsar el botón de ENTER del PC para
ejecutar el programa downmcu. Resumiendo:

1. Teclearen la línea de comandos: downmcu miprograma-com1

2. Pulsar el botón de resetde la BT6811

3. Pulsar el botón de ENTER del PC para ejecutar el programa downmcu.

Mipr ogramaes el nombre del programa con extensión s19,es decir ya compilado, que se
quiere enviar a la BT6811. Si la conexión seha realizado por el puerto serie dos, habrá que
poner como parámetro -com2.

Otros programas que pueden utilizarse en modo entrenador son el mcbooty el ctdialog.
El primer o consiste en un pequeño terminal que manda por el puerto serie todo lo que
recibedel teclado y envía a la pantalla todo lo que recibepor el puerto serie. La forma de
utilizarlo esla siguiente:

1. Teclearen la línea de comandos: mcboot

2. Pulsar F4para seleccionarel puerto serie adecuado.

3. Finalmente pulsar F1 para cargar programas en la RAM interna. En el procesode
carga habrá que hacer lo mismo que con el downmcu,pulsar el resetde la BT6811 y
luego pulsar ENTER para enviar el programa al micro.

Mediante el mcboot,una vez cargada la aplicación deseadaen la BT6811, se devuelve el
control al terminal del mcboot.Por esosi el programa enviado a la BT6811 permite recibir
carácterespor el puerto seriesele pueden enviar con tan solo pulsar las teclasen el mcboot.
A su vez si la BT6811 envía datos por el puerto serie estosaparecerán en la pantalla del
mcboot.



A.6. UTILIZACIÓN DE LA BT6811 149

Por último, en el modo entrenador hay otro programa que adquiere mucha importan-
cia a la hora de buscar aplicaciones autónomas. Esteprograma esel ctdialog, que como se
acabade decir a pesarde ejecutarseen modo entrenador esel que permite, entre otras fun-
ciones,dejar grabados los programas permanentemente en la EEPROM interna, de forma
que seobtienen aplicaciones autónomas del PC.

Paraejecutarlo en MSDOShay que hacer lo siguiente:

1. downmcu ctserver -com1

2. ctdialog -com1

Siel puerto serieutilizado esel segundo hay que poner -com2y tener presentesiempreque
la BT6811 no dispone de resetsoftware por lo que habrá que hacerlo manual al utilizar el
downmcu. En el ctdialogno esnecesarioya que opera sobre el servidor ctserverpreviamente
cargado en el microcontrolador.

Para ejecutarlo en Linux no hace falta el paso primer o, con poner ctdialog -com1 bas-
tará, aunque ahora sí es necesario pulsar el botón de reset de la BT6811 justo antes de
pulsar el ENTER. Una vez que seha cargado el programa apareceráen pantalla lo que se
muestra en la �gura A.5. A partir de esemomento se puede grabar un programa en la
memoria EEPROM2 mediante la orden siguiente:

1. ms 10350 (Sólo para el micro E2 y sólo en MSDOS)

2. eeprom mipr ograma

Despuésde terminar la ejecuciónde la orden (apareceun mensajede OK), el microcontro-
lador de la BT6811 tendrá grabado en su memoria EEPROM el programa. Esta memoria
no es volátil, permaneciendo inalterado su contenido independientemente del estado de
las baterías. Esta característicahaceposible las aplicaciones autónomas que se describen
en los siguientes apartados.

A.6.2 BT6811 en modo autónomo aislado

Al decir que la BT6811 esta funcionando en modo autónomo se hace referencia a que el
programa de la aplicación ya se encuentra grabado en la EEPROM interna del microcon-
trolador, por lo que no esnecesarioutilizar el PC para descargar los programas y probar.
Esdecir no hacefalta utilizar el adaptador que anteshabía sido necesarioincorporar.

¿ Qué aplicaciones se pueden hacer con una BT6811 ?. La respuesta una vez más:
depende de la imaginación del lector, pero se pueden citar ejemplos de cómo dotar de

2La memoria EEPROM actua como una EPROM pero que seborra eléctricamente. La particularidad de
una EPROM esque sus contenidos segraban electricamente pero seborran con luz ultravioleta, por esose
mantienen independientemente de si hay o no tensión eléctrica. Con la EEPROM seda un paso más ya que
sesustituye la luz ultravioleta por unos pulsos de tensión especiales,de forma que el procesode borrado y
escritura sepuede realizar sin necesidadde agentesexternos y en un tiempo menor. Estamemoria haceque
la familia de microcontroladoresMC68HC11 haya tenido una aceptación muy grande en la industria, pues
de una manera muy fácil permite programar aplicaciones que no necesiten de mecanismos externos para
reteneren memoria el programa, considerándoseasímicros autónomos o preprogramados.

Figura A.5: Programa CTDIALOG bajo Linux.

movimiento a un robot o a una maqueta. Con una BT6811 se podrían mover alternativa-
mente cuatro motores,ocho luces y tal vez emitir alguna señal acústicade vez en cuando.
Serecuerda que el motivo principal del desarrollo de estatarjeta esel poder controlar hasta
cuatro servomecanismosFutaba para la realización de robots autónomos. Además tiene
su�ciente capacidad para gestionar otros recursoscomo los anteriormente citados : leds,
sonidos, bumpers, etc.

Las ventajasde usar la tarjeta en estemodo son:

1. Tiene un tamaño reducido y un coste inferior a la CT6811 ya que sólo utiliza los
componentes impr escindibles, el resto los elimina.

2. Ya viene preparada para conectarcuatro servomecanismosFutaba.

El lector puede considerar que estascaracterísticasno son su�cientes para justi�car la uti-
lización de la BT6811 en lugar de la CT6811 y en parte lleva razón. Serecuerda que para
programarla senecesitauna CT6811 para realizar todo lo descrito en el punto anterior. Y si
ya sedispone de una CT6811,esmás fácil construirse el adaptador para los motoresque la
propia BT6811. Entonces,¿dónde está la verdadera utilidad ?. La respuestaseencuentra
en las redesde control distribuido, en donde conceptoscomo la modularidad, el costey el
tamaño toman bastante importancia. En el apartado siguiente se explicarán ejemplos de
esto.

Para con�gurar la BT6811 en modo autónomo (independientemente de si esaislado o
en red) sepueden emplear dos métodos. El primer o seusacuando el microcontrolador que
lleva la tarjeta es un MC68HC11A1 y el segundo se emplea cuando el microcontrolador
es un MC68HC11E2. La difer encia entre ambos se encuentra en que en el primer caso
el microcontrolador tiene situada la EEPROM en la dir ección $B600y la única forma de
accedera ella nada másactivar la placa esmediante el modo bootstrapy la colocaciónde un
jumper. Por el contrario en el segundo casoel microcontrolador tiene situada la EEPROM
en el rango $F800- $FFFFpor lo que el vector de resetpuede programarse en ella y por
ello sehabilita el modo de arranque single-chip. Mediante estemodo nada más activar la
tarjeta seejecutael programa indicado por el vectordereset.

Con�guración del modo autónomo para el microcontrolador 68HC11A1:

1. Jumper JP1puesto para con�gurar el modo bootstrap.

2. Jumper JP2puesto para permitir el salto a la EEPROM y ejecutar así el programa
grabado en ella.

3. Los demás jumpers no tienen importancia en la seleccióndel modo aunque sí en el
restode funcionalidades.

Con�guración del modo autónomo para el microcontrolador 68HC11E2:

1. Jumper JP1quitado para con�gurar el modo single-chip.

2. Jumper JP2quitado.

3. Vector de resetgrabado en la EEPROM y apuntando al inicio del programa.

4. El resto de jumpers no tienen importancia en la seleccióndel modo aunque sí en el
restode funcionalidades.

Para terminar el apartado seexplicará mediante un ejemplo cómo obtener una aplicación
autónoma. Lo importante del ejemplo no es la funcionalidad del mismo sino el proceso
que se va a seguir para tener el sistema funcionando, sobre todo porque en el apartado
siguiente todo lo explicado aquí sedará por aprendido. Además, seva a considerar sólo el
segundo caso,esdecir cuando el microcontrolador de la BT6811 esun MC68HC11E2.

El ejemplo pretende que en cuanto seenchufe la BT6811, empiecea parpadear su LED.
El código del programa seríael siguiente:

* Programa que hace parpadear el LED de la BT6811

PORTB EQU $04

ORG$F800 * Micro E2

inicio * Inicio del programa
LDX #$1000

bucle
LDAA PORTB,X
EORA #$10
STAA PORTB,X
BSR pausa
BRA bucle

pausa
LDY #$FFFF

sigue
DEY
CPY #$0
BNE sigue
RTS

* Vector de interrupción del reset

ORG$FFFE
v_reset FDB #inicio

Lo importante del programa esver cómo seha con�gurado el vector de reset(últimas dos
líneas)apuntando al inicio del programa, de manera que cuando searranque en single-chip
seempiecea ejecutar la instrucción situada en la dir ección guardada en el vector. Esains-
trucción tiene que coincidir con el inicio del programa. Si no secon�gura correctamente,

el programa empezará a ejecutarsepor cualquier lado siendo impr edecible su compor-
tamiento. En casode utilizar un microcontrolador MC68HC11A1 estevector no seutiliza,
por lo que no hará falta con�gurarlo y en cambio habrá que considerar que cada vez que
se haga un resety se tenga la placa con�gurada correctamente se empezará a ejecutar la
instrucción del programa que seencuentra en la línea $B600.

El programa anterior hay que compilarlo y mediante el ctdialoggrabarlo en la EEPROM
interna. La forma de realizar estoúltimo seha explicado en el apartado anterior.

1. Paracompilar : as11 mipr ograma.asm

2. Ejecutar el ctdialog: > ctdialog -com13

3. Paragrabar : eeprom mipr ograma

Una vez grabado el programa se con�gura la BT6811 y si todo ha ido bien, cada vez que
se pulse el reseto se enchufe la tarjeta el LED empezará a parpadear. A continuación se
explicará el modo autónomo en red que, como ya seha dicho, escon el que seobtiene la
máxima utilidad de la BT6811.

A.6.3 BT6811 en modo autónomo en red

La BT6811 como sistema de control aislado tiene unas ventajas limitadas, pero cuando se
utilizan varias tarjetas para formar un sistema distribuido de control éstasaumentan. Ya
no sólo esimportante el tamaño, sino el precioy la funcionalidad de cadaunidad. Suponga
por un momento que deseaconstruir un robot que tenga cuatro servomecanismosy que
seaautónomo. Mediante una BT6811 aislada lo puede hacer, pero ¿quéocurre si más tarde
deseaampliar el robot ?,esdecir, ponerle másmotores.Esevidente que necesitaráampliar
el hardwar e y lo más probable esque utilice otra BT6811. Ahora bien, el software hay que
cambiarlo, ahora hay dos tarjetas en lugar de una: ¿cómorepartir el programa?, ¿ cómo
hacerlo en futuras ampliaciones?. Son preguntas que sirven para demostrar la utilidad
de la red de control. Si cada vez que se intr oduce un módulo hay que recti�car todos los
programas, algo se está haciendo mal. Además, cuando se construye un robot es muy
frecuente que a medida que se van obteniendo buenos resultados se vayan adquiriendo
nuevos retos. Y estáclaro que si sesigue por el camino de modi�car todo cada vez que se
intr oduce un nuevo elemento, mal camino seha escogido.

¿Cómo se puede evitar lo anterior ?. La respuestaestáen la modularidad. La mayo-
ría de los proyectossepueden subdividir en módulos más o menos independientes y con
funcionalidades parecidas o totalmente dispares. De forma que para retocar un módulo
no haga falta atacara los otros. Además si sedecidiesea añadir más, lo más seguro esque
no se necesite reprogramar todo el sistema, o por lo menos de eso se trata. Todos estos
módulos no deben actuar por su cuenta,sino que tienen que guardar un sincronismo entre
ellos, deben intercambiar información o por lo menos tener uno maestro que controle a los
demás. Seríacomo construir una plataforma móvil partiendo de bloques ya establecidos,
es decir, utilizando un módulo motriz, un módulo sensorial, un módulo actuador y para

3Considerar que senecesitael adaptador, lo del resetsoftware y la difer encia entre el método en linux y
msdos. esdecir todo lo explicado en 6.6.1.

Figura A.6: Representaciónde la red de BT6811.

tener todo comunicado un módulo maestro que gestione los anteriores. Si una vez desa-
rrollado hay que añadir algún nuevo elemento, tan sólo hay que añadirlo a la estructura
y modi�car el módulo de gestión, los demás deberían permanecer inalterados. Para lo-
grar esto serequiere una buena plani�cación del proyecto y una descripción buena de los
interfaces entre módulos.

Debido a que los ejemplos suelen ser más útiles que las palabras expondremos uno.
Apr ovechando el LED de la BT6811 sediseñará una red de control que permita desde un
módulo maestro encendery apagar cadaLED de cadaBT6811.

Lo primer o que hay que hacer es pensar en la aplicación desde un punto de vista ge-
neral: el esquema de la �gura A.6 servirá de ayuda. Seutilizará el PC para mandar las
órdenes de encender y apagar el LED de cada BT6811. Habrá un módulo maestro que
recibirá por el puerto serie los datos del PC y después los enviará por la red de control.
A esta red están conectadastodas las BT6811 que uno quiera y a pesar de que todas ellas
recibirán la misma orden, no habrá con�icto. Mediante un identi�cador cada BT6811 po-
drá saber si la orden va o no dirigida a ella y en el casoa�rmativo procederáa encender
o apagar su LED. En esteejemplo todos los módulos van a tener la misma función, pero
puede haber otros casosen los que no. En estos,habrá que estudiar bien qué órdenes se
van a transmitir por la red. Cuanto mejor seespeci�quen éstas,más detallado setendrá el
protocolo �nal de la red y menos modi�caciones setendrán que hacer a largo plazo.

En esteejemplo todos los módulos BT6811 tendrán la misma función: Encender y apa-
gar su LED cuando selo indique el módulo maestro. Con estadecisión sepuede especi�car
el protocolo de la red o, lo que es lo mismo, el formato de las tramas y el procedimiento
para enviarlas por la red.

Cada trama tendrá dos bytes, el primer o identi�cará la BT6811 a la cual va dirigida la
orden y el segundo será la propia orden, es decir, encender o apagar el LED, ver la tabla
A.2. El identi�cador Id seráun número entre 1 y 9, y el código de orden, n_orden, seráel
carácter 'a' para encenderel LED y el carácter 's' para apagarlo.

Dicho lo anterior seva a proceder a programar el módulo BT6811. Con hacer un pro-

Identi�cador de destino orden
1 Byte 1 Byte

Id n_orden

Tabla A.2: Especi�caciones de la trama de control.

grama general bastaráy para añadir másmódulos tan sólo habrá que ir cambiando el valor
de una etiqueta, en concreto la indicada en el programa por M_ID . El código del programa
sedetalla a continuación.

* ............................... .... ..... ..
* . BTgen. Version 1.0 .
* ............................... .... ..... ..
* . Modulo general de la red de BT6811 .
* ............................... .... ..... ..

**************************** ***** **** ***** **
* Identificador de la BT6811 *
* Incrementar una unidad en cada .nueva BT *
**************************** ***** **** ***** **

M_ID equ '1'

**************************** ***** **** ***** **
* Registros y Puertos del 68hc11 *
**************************** ***** **** ***** **

* Puertos del 68hc11

PORTB equ $04

* Registros del SPI

PORTD EQU $08
DDRD EQU $09
SPCR EQU $28
SPDR EQU $2A
SPSR EQU $29

**************************** ***** **** ***** **
* Máscaras de acceso *
**************************** ***** **** ***** **

LED equ $10 ; Bit donde se encuentra el LED

**************************** ***** **** ***** **
* ------------------------------- ---- ----- *
* | PROGRAMAPRINCIPAL | *
* ------------------------------- ---- ----- *
**************************** ***** **** ***** **

ORG$F800 * Programa para un E2

**************************** ***** **** ***** **
* Inicialización del programa *
**************************** ***** **** ***** **

inicializar
LDS #$FF * puntero de pila
LDX #$1000 * acceder a registros

*--- Configuración del SPI (como esclavo)

LDAA #$3C * SS de entrada
STAA DDRD,X

LDAA #$40 * Colector cerrado
STAA SPCR,X * Modo esclavo el SPI

**************************** ***** **** ***** **
* BUCLE PRINCIPAL *
**************************** ***** **** ***** **

inicio
BSR recibir_spi
CMPA#M_ID * verifica el ID
BEQ analizar * SI->lee orden
BSR recibir_spi * .NO->no hagasñada
BRA inicio

analizar BSR recibir_spi * lee la orden
CMPA#'a' * ¿ Es a ?
BEQ led_on * SI -> enciende led
CMPA#'s' * ¿ Es b ?
BEQ led_off * SI -> apaga led
BRA inicio

**************************** ***** **** ***** ****
* SUBRUTINAS DEL PROGRAMA *
**************************** ***** **** ***** ****

*........................... ....
*. Rutina enciende el LED .
*........................... ....

led_on
BSET PORTB,X LED
BRA inicio

*........................... ....
*. Rutina apaga el LED .
*........................... ....

led_off
BCLR PORTB,X LED
BRA inicio

* ............................... .... ..... .... .
* . Rutina que recibe un dato por el SPI .
* . La rutina espera hasta recibir el dato .
* . Entradas:Ninguna .
* . Salidas: El acumulador A contiene el dato .
* ............................... .... ..... .... .

recibir_spi
espera BRCLR SPSR,X $80 espera * espera dato

LDAA SPDR,X
RTS

**************************** **
* vectores de interrupcion *
**************************** **

ORG$FFFE * vector del reset
v_reset FDB #inicializar

END

El programa no escomplicado: el bucle principal lo único que haceesesperar a recibir el
primer byte de cada trama enviada por el bus de control (formado por el SPI). Este byte
(Id ) identi�ca a la tarjeta a la que seha enviado la orden, por eso lo siguiente que haceel
bucle principal es comparar el byte con su identi�cador (M_ID ). Si no coinciden ambos
el bucle lee el siguiente byte de la trama, pero no lo utiliza, vuelve al bucle principal a
esperar otro principio de trama. Por el contrario si ambos identi�cador es son iguales, el
bucle leeel siguiente byte, lo analiza y en función de su contenido enciendeo apagael LED

Figura A.7: Cable para conectar los distintos módulos entre sí.

de la BT6811. Una vez terminado estovuelve al bucle principal para seguir esperandouna
nueva trama. Tal vez la parte más complicada es la de recibir los datos por el bus SPI ya
que para ello hay que con�gurarlo de una manera especial, en concreto en los módulos
esclavoshay que programarlo con los parámetros siguientes: colector cerrado, señalSSde
entrada y modo esclavo. Todo esto sehaceen la inicialización del programa. Por último,
como el microcontrolador que seestáutilizando esel MC68HC11E2pondr emos el vector
de resetapuntando al principio del programa para poder arrancar en modo single-chip4.

El lector puede preguntarse cómo se conectan los módulos entre sí, la respuesta es
sencilla, o se construye un cable que una los pines MOSI, SS,CLK y GND de todos los
módulos incluyendo el maestro, o seconstruye un cable tipo bus con tantos conectoresde
salida como módulos tenga la red. Esta última opción es la recomendada en el ejemplo,
aunque sepodría tomar la precauciónde cortar los cablescorrespondientesal puerto serie
(PD1 y PD0) en el enganche al módulo maestro, de esa forma se evitarán con�ictos con
los puertos serie de los módulos esclavos.En la �gura A.7 serepresentacómo construirse
el cable de conexión del bus SPI. Por último, para conectar el PC al módulo esclavo se
utilizará el cable serie proporcionado en la CT6811. La alimentación del sistema se intr o-
ducirá, por ejemplo, por el jack de la CT6811 y sedistribuirá por toda la red uniendo con

4En el apartado 'A.2: modos de funcionamiento' y en el 'A.6.2: funcionamiento en modo autónomo
aislado' hay más información sobre el modo de arranque single-chip.

un cable todas las salidas positivas de las clemasV_TTL de cadaBT6811 (ver �gura A.4).
Ya setiene construida la red y el programa general de los módulos esclavos.Ahora se

tiene que grabar ésteen cada módulo, para lo que seutiliza toda la teoría explicada en el
apartado anterior. Al �nal se con�gurará cada BT6811 en modo autónomo aislado, pero
con un par de añadidos por utilizar la red SPI,en concreto:

1. Jumper JP1quitado para con�gurar el modo single-chip.

2. Jumper JP2quitado.

3. Jumper JP4quitado ya que en la red seva a colocar una CT6811.

4. Jumper JP6puesto ya que el control por el SPIva a utilizar la señalSS.

5. El restode jumpers no tienen importancia en la aplicación.

Al grabar el programa en cada BT6811 hay que acordarse de modi�car la etiqueta M_ID
que identi�ca a cada módulo. Seempieza dándole el valor uno al primer módulo y seva
incrementando a medida que se van grabando más módulos. En casode dejar el mismo
valor para todos, ocurrirá que todos los LEDs seencenderáno apagarán a la vez.

Ahora queda programar el módulo maestro, que como ya seha dicho esuna CT6811.
Como va a estar conectado al PC se puede elegir entre tener grabado el programa en la
memoria EEPROMo volcarlo cadavez que seutilice la aplicación. Parafacilitar lascosasse
hará de estasegunda forma, esdecir sedejará con�gurada en modo entrenador5. Seusará
el MCBOOT para enviar el programa maestro a la CT6811 y luego tecleando una serie de
órdenes severá como el LED de los módulos esclavoscambia de estado. A continuación
sedetalla el código del programa maestro.

**************************** ***** **** ***** ***
* Programa para el modulo maestro *
**************************** ***** **** ***** ***
* El programa recibe ordenes por el puerto*
* serie, las interpreta y manda las tramas *
* adecuadas por la red_SPI para que le *
* llegue la orden a la BT6811 adecuada. *
**************************** ***** **** ***** ***

* --- Puertos de entrada y salida

PORTA EQU $0

* --- Registros del SCI

BAUD EQU $2B
5Para ver la con�guración en modo 'entrenador' recurrir al 'manual deusuariodela CT6811'. Este se en-

cuentra disponible en la WEB de Micr obótica S.L.[10].

SCCR1 EQU $2C
SCCR2 EQU $2D
SCSR EQU $2E
SCDR EQU $2F

* --- Registros del SPI

PORTD EQU $08
DDRD EQU $09
SPCR EQU $28
SPDR EQU $2A
SPSR EQU $29

ORG$F800 * micro E2

inicio LDS #$FF * pila
LDX #$1000 * registros

*---- Configuracion del SCI

wait BRCLR SCSR,X $40 wait

LDAA #$22 * Poner 22 para 7812 baudios
STAA BAUD,X * Poner 30 para 9600 baudios
LDAA #$0
STAA SCCR1,X * 8 bits de datos
LDAA #$0C *.No usar interrupciones del SCI
STAA SCCR2,X * Activar receptor y transmisor

*---- Configuracion del SPI (como maestro)

LDAA #$3C
STAA DDRD,X
LDAA #$50 * Colector cerrado
STAA SPCR,X * Activa en modo maestro el SPI

**************************** ***
* BUCLE PRINCIPAL *
**************************** ***

bucle_prin
JSR leer_car * leo carácter por el puerto se-

rie
JSR enviar_car * hace eco del caracter
JSR enviar_spi * mando carácter al bus spi



A.6. UTILIZACIÓN DE LA BT6811 161

BRA bucle_prin

*........................... ..... .
*. Rutina que recibe un caracter .
*. por el puerto serie .
*........................... ..... .

leer_car
BRCLR SCSR,X $20 leer_car
LDAA SCDR,X
RTS

*........................... ..... .
*. Rutina que envia un caracter .
*. por el puerto serie .
*........................... ..... .

enviar_car
BRCLR SCSR,X $80 enviar_car
STAA SCDR,X
RTS

*........................... ..... .... ...
*. Rutina que envia un dato por el SPI .
*........................... ..... .... ...

enviar_spi
LDAB PORTD,X
ANDB #$DF
STAB PORTD,X * activarse al esclavo

STAA SPDR,X * introduzco el dato a mandar
espera BRCLR SPSR,X $80 espera

LDAB PORTD,X
ORB #$20
STAB PORTD,X * Desactivamos al esclavo

RTS

**************************** *****
* Vector de interrupcion *
**************************** *****

ORG$FFFE
v_reset FDB #inicio

END

El programa maestro tampoco esdifícil de comprender y sepodría reducir más, ya que al
utilizar el modo entrenador el puerto serieseauto con�gura, igual que la pila, por lo que se
podían haber ahorrado todas esaslíneasde con�guración. Las líneas de con�guración del
SPI vuelven a ser claves, aunque ahora se ha con�gurado como colector cerrado y como
maestro.

Despuésde la con�guración se ejecuta el bucle principal que lo único que hacees es-
perar a recibir datos del PC. En cuanto recibe uno hace eco y lo envía por el bus SPI a
los módulos esclavos. Por lo tanto si con el MCBOOT se teclea '1' y luego 'a' se debería
encender el LED del módulo esclavo cuyo identi�cador M_ID es igual a uno. Si se hace
lo mismo pero tecleando un '2' debería pasar lo mismo pero en el módulo esclavo cuyo
M_ID fuera dos. Paraapagar el LED de la BT6811 hay que pulsar primer o el '1' y luego 's'.

Si una vez terminada la aplicación sequiere añadir un nuevo módulo habrá que hacer
lo siguiente:

1. Seincrementa el valor M_ID del programa de los módulos esclavos(BT6811).

2. Segraba dicho programa en el nuevo módulo y secon�gura en modo autónomo en
red.

3. Seconectael módulo al bus SPIy a la alimentación de 5 voltios.

4. El restode módulos sedejan como están.

Tal como sedijo al principio del ejemplo al ser un sistemamodular no ha hecho falta tocar
ninguno de los módulos ya existentespara ampliar el sistema. El lector puede considerar
que esteejemplo esmuy sencillo, pero en la realidad hay muchos sistemasde control que
seadaptan al ejemplo. De todas formas la BT6811 todavía admite más complejidad en la
red. Por ejemplo, en lugar de usar la señal SScomún para toda la red sepuede indepen-
dizar (quitar el jumper JP6).En su lugar seconectaráuna salida del módulo maestro, como
puede ser algún pin del puerto B. Haciendo esto se consigue una comunicación de ida y
vuelta entre el maestro y el esclavo, de forma que se puede intercambiar información en
ambos sentidos. Además, en lugar de ser la trama la que identi�ca el destino es el pro-
pio maestro el que lo hace. En este casoal incluir un nuevo módulo hay que retocar el
módulo maestro para que lo reconozca,pero por el contrario, estenuevo módulo puede
tener una funcionalidad distinta sin que afectea la red. Como podrá apreciar el lector las
posibilidades son muchas y cuanto más practique más irán surgiendo. Para adquirir ex-
periencia serecomienda leerseel capítulo relacionado con el SPIdel libr o Microcontrolador
MC68hc11: Fundamentos,recursosy programación[9].


